F07PJF (SSPTRI/DSPTRI) – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

F07PJF (SSPTRI/DSPTRI) computes the inverse of a real symmetric indefinite matrix A, where A has been factorized by F07PDF (SSPTRF/DSPTRF), using packed storage.

2 Specification

```
SUBROUTINE F07PJF(UPLO, N, AP, IPIV, WORK, INFO)
ENTRY      ssptri(UPLO, N, AP, IPIV, WORK, INFO)
INTEGER            N, IPIV(*), INFO
real               AP(*), WORK(*)
CHARACTER*1        UPLO
```

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

To compute the inverse of a real symmetric indefinite matrix A, this routine must be preceded by a call to F07PDF (SSPTRF/DSPTRF), which computes the Bunch–Kaufman factorization of A using packed storage.

If UPLO = 'U', $A = PUDU^TP^T$ and A^{-1} is computed by solving $U^TP^TXPU = D^{-1}$.

If UPLO = 'L', $A = PLDL^TP^T$ and A^{-1} is computed by solving $L^TP^TXPL = D^{-1}$.

4 References

[1] Du Croz J J and Higham N J (1992) Stability of methods for matrix inversion *IMA J. Numer. Anal.* **12** 1–19

5 Parameters

1: UPLO — CHARACTER*1 *Input*

On entry: indicates how A has been factorized as follows:

if UPLO = 'U', then $A = PUDU^TP^T$, where U is upper triangular;
if UPLO = 'L', then $A = PLDL^TP^T$, where L is lower triangular.

Constraint: UPLO = 'U' or 'L'.

2: N — INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N ≥ 0.

3: AP(*) — ***real*** array *Input/Output*

Note: the dimension of the array AP must be at least max(1,N*(N+1)/2).

On entry: details of the factorization of A stored in packed form, as returned by F07PDF (SSPTRF/DSPTRF).

On exit: the factorization is overwritten by the n by n symmetric matrix A^{-1} stored in packed form. More precisely, the (i,j)th element of A^{-1} is stored in AP$(i + j(j-1)/2)$ for $i \leq j$ if UPLO = 'U', and in AP$(i + (2n-j)(j-1)/2)$ for $i \geq j$ if UPLO = 'L'.

4: IPIV(∗) — INTEGER array *Input*

Note: the dimension of the array IPIV must be at least max(1,N).

On entry: details of the interchanges and the block structure of D, as returned by F07PDF (SSPTRF/DSPTRF).

5: WORK(∗) — ***real*** array *Workspace*

Note: the dimension of the array WORK must be at least max(1,N).

6: INFO — INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO $= -i$, the ith parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO $= i$, d_{ii} is exactly zero; D is singular and the inverse of A cannot be computed.

7 Accuracy

The computed inverse X satisfies a bound of the form

$$\begin{array}{ll} |DU^TP^TXPU - I| \leq c(n)\epsilon(|D||U^T|P^T|X|P|U| + |D||D^{-1}|) & \text{if UPLO = 'U', or} \\ |DL^TP^TXPL - I| \leq c(n)\epsilon(|D||L^T|P^T|X|P|L| + |D||D^{-1}|) & \text{if UPLO = 'L',} \end{array}$$

where $c(n)$ is a modest linear function of n, and ϵ is the ***machine precision***.

8 Further Comments

The total number of floating-point operations is approximately $\frac{2}{3}n^3$.

The complex analogues of this routine are F07PWF (CHPTRI/ZHPTRI) for Hermitian matrices and F07QWF (CSPTRI/ZSPTRI) for symmetric matrices.

9 Example

To compute the inverse of the matrix A, where

$$A = \begin{pmatrix} 2.07 & 3.87 & 4.20 & -1.15 \\ 3.87 & -0.21 & 1.87 & 0.63 \\ 4.20 & 1.87 & 1.15 & 2.06 \\ -1.15 & 0.63 & 2.06 & -1.81 \end{pmatrix}.$$

Here A is symmetric indefinite, stored in packed form, and must first be factorized by F07PDF (SSPTRF/DSPTRF).

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F07PJF Example Program Text
*     Mark 15 Release. NAG Copyright 1991.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX
      PARAMETER        (NMAX=8)
*     .. Local Scalars ..
      INTEGER          I, IFAIL, INFO, J, N
      CHARACTER        UPLO
*     .. Local Arrays ..
      real             AP(NMAX*(NMAX+1)/2), WORK(NMAX)
      INTEGER          IPIV(NMAX)
*     .. External Subroutines ..
      EXTERNAL         ssptrf, ssptri, X04CCF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F07PJF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*        Read A from data file
*
         READ (NIN,*) UPLO
         IF (UPLO.EQ.'U') THEN
            READ (NIN,*) ((AP(I+J*(J-1)/2),J=I,N),I=1,N)
         ELSE IF (UPLO.EQ.'L') THEN
            READ (NIN,*) ((AP(I+(2*N-J)*(J-1)/2),J=1,I),I=1,N)
         END IF
*
*        Factorize A
*
         CALL ssptrf(UPLO,N,AP,IPIV,INFO)
*
         WRITE (NOUT,*)
         IF (INFO.EQ.0) THEN
*
*           Compute inverse of A
*
            CALL ssptri(UPLO,N,AP,IPIV,WORK,INFO)
*
*           Print inverse
*
            IFAIL = 0
*
            CALL X04CCF(UPLO,'Nonunit',N,AP,'Inverse',IFAIL)
*
         ELSE
            WRITE (NOUT,*) 'The factor D is singular'
         END IF
      END IF
      STOP
```

```
*
      END
```

9.2 Program Data

```
F07PJF Example Program Data
  4                           :Value of N
  'L'                         :Value of UPLO
  2.07
  3.87  -0.21
  4.20   1.87   1.15
 -1.15   0.63   2.06  -1.81   :End of matrix A
```

9.3 Program Results

```
 F07PJF Example Program Results

 Inverse
             1          2          3          4
 1      0.7485
 2      0.5221    -0.1605
 3     -1.0058    -0.3131     1.3501
 4     -1.4386    -0.7440     2.0667     2.4547
```